FOIPOP Request 2022-01337-HEA

Request:

In the Nova Scotia Monthly COVID-19 Epidemiologic Summary dated August 9, 2022

Found here: https://novascotia.ca/coronavirus/alerts-notices/#epidemiologic-summaries

It states that:

"Age and lack of vaccination continue to be associated with severe outcomes:

>>>"Unvaccinated Nova Scotians were hospitalized at nearly four times the rate and died more than three times the rate as those with three or more doses."<<<

I would like to receive all records that show the step by step mathematical calculation with supportive documentation and explanations - the calculation that was used to determine, support and justify this statement, that:

"Unvaccinated Nova Scotians were hospitalized at nearly four times the rate and died more than three times the rate as those with three or more doses."

In essence, the "record" that shows how was this was determined by the Public Health Branch of the Nova Scotia Department of Health and Wellness.

(Date Range for Record Search: From 12/31/2021 To 08/23/2022)

Response:

The statement being queried is based on a relative risk calculation using data in Table 3 of *Nova Scotia's Monthly COVID-19 Epidemiologic Summary*. A relative risk is a common way epidemiologists compare outcomes in two different groups of people. It divides the "risk" of an outcome in people in one group to the "risk' of an outcome in people in another group. Note: the term 'risk' is nuanced but can be thought of as an umbrella term to capture "rate", "incidence", "incidence rate", "incidence proportion". Risk, in its most basic form, is the number of new events in a population divided by the population at risk for the event in a population.

$$Risk = \frac{new\ events}{population\ at\ risk}$$

In this instance, the "risks" of interest would be the rate of hospitalizations (in unvaccinated and in people with 3+ doses) and the rate of death (in unvaccinated and in people with 3+ doses). The basic calculations for these are shown below.

Rate of hospitalization
$Rate_{unvaccinated} = \frac{\#\ hospitalizations\ in\ unvaccinated}{person-time\ contributed\ by\ people\ unvaccinated} = 346.8$
$Rate_{3+doses} = \frac{\#\ hospitalizations\ in\ 3+doses}{person-time\ contributed\ by\ people\ with\ 3+doses} = 310.1$
Rate of death
$Rate_{unvaccinated} = \frac{\#\ deaths\ in\ unvaccinated}{person-time\ contributed\ by\ people\ unvaccinated} = 48.9$
$Rate_{3+doses} = \frac{\#\ deaths\ in\ 3+doses}{person-time\ contributed\ by\ people\ with\ 3+doses} = 66.7$

Note, that the person time is dynamic and changes as vaccine status changes. We require computer programs to calculate the person-time denominator. These computer programs use the health information from administrative health data. To determine person time, the following information needs to be known

- How many people are in the population
- How many doses of vaccine they received
- When they received that vaccine

This information is used to calculate the denominator, person-time, which is in turn, used to calculate the rates/risks.

These risks are then adjusted for age. Age is a strong predictor of both death and vaccination. Older people are more likely to die of COVID-19 and they are also more likely to have received 3 or more doses. Adjusting for age is a very common approach used by epidemiologists that allows for more fair comparisons between two groups of people that may have different ages.

Once the age-adjusted rates are calculated, "relative risk' of the outcomes – hospitalization and death – people who are vaccinated with 3 or more doses to people who have not been vaccinated can be calculated.

$$RR = \frac{age-adjusted\ rate\ unvaccinated}{age-adjusted\ rate\ 3+doses}$$

Relative Risk of hospitalization	Relative Risk of death
$RR = \frac{756.2}{204.2} = 3.7$	$RR = \frac{144.6}{41.4} = 3.5$

From these calculations, we see the "Unvaccinated Nova Scotian's were hospitalized at 3.7 times (i.e. nearly 4 times the rate) and died at 3.5 (i.e. more than 3 times the rate) as those with three or more doses.

Note: the epidemiology team uses a computer program (SAS) to calculate rates, age-adjusted rates, and relative risks. A copy of SAS computer code is included in this response.

Table 3: Age-adjusted hospitalization* and death rates by vaccine status, March 1, 2022 to July 31, 2022 (Wave 6)

Vaccination Status	Number	Crude Rate per 100k Person-Years	Age-Adjusted Rate per 100k Person-Years
Hospitalizations			
Unvaccinated	220	346.8	756.2
2 Doses	252	201.0	412.2
3+ Doses	662	310.7	204.2
Deaths			
Unvaccinated	31	48.9	144.6
2 Doses	39	31.1	94.1
3+ Doses	142	66.7	41.4

Data sources: Hospitalizations - PPHLN, Meditech and STAR; Deaths – Panorama; Denominator - MSI Eligible Nova Scotians

Notes:

- *Hospitalizations for individuals missing age are excluded from the analysis (counts, crude rates, age-adjusted rates, risk reduction)
- A person is considered unvaccinated when they have zero doses of any COVID-19 vaccine
- A person is considered to have two doses 14 or more days after the second dose of any vaccine OR 14 or more days after one dose of Johnson & Johnson vaccine OR are within 14 days of receiving a third dose of any COVID-19 vaccine
- A person is considered to have three or more doses 14 or more days after a third dose of any COVID-19 vaccine
- Due to small number of events occurring among those just one dose, that group is not included in this analysis.

- To more accurately estimate vaccine coverage in Nova Scotia, the province is switching from using Statistics Canada population estimates to the number of Nova Scotians eligible for MSI (the province's health insurance card) as the denominator for estimating both per capita rates and vaccine coverage.

```
/*age-adjusted rates*/

/*deaths*/
data library.ratebase;
      merge pano_nsh_summary combdenoms;
      by wavetxt agegrp;
      if dosegrp = 0 then vaccpt = c_vaccpt0;
      if dosegrp = 1 then vaccpt = c_vaccpt1;
      if dosegrp = 2 then vaccpt = c_vaccpt2;
      if dosegrp = 3 then vaccpt = c_vaccpt3;
run;

data deathrate;
      set library.ratebase;
      keep wavetxt agegrp dosegrp panodeathcount pt;
      where wavetxt = "&currentwave" and agegrp in('<50 years', '50-69
years', '70+ years');
      pt = vaccpt;
run;

proc sort data = deathrate; by dosegrp agegrp; run;

data provrate;
      set library.ratebase;
      keep wavetxt agegrp pt;
      where wavetxt = "&currentwave" and agegrp in('<50 years', '50-69
years', '70+ years');
      pt = agept;
run;

proc sort data = provrate nodupkey; by agegrp; run;

ods graphics on;
ods select StdRate;
proc stdrate data=deathrate
      refdata=provrate
      method=direct
      stat=rate(mult=100000)
      plots=none;
      by dosegrp;
      population  event=panodeathcount total=pt;
      reference total=pt;
      strata agegrp;
      ods output StdRate=deathadj;
run;
ods graphics off;

data deathadj (keep=outcome dosegrp events cr ar);
      set deathadj;
      outcome = "Deaths";
      cr = round(cruderate*365.5,0.1);
      ar = round(stdrate*365.5,0.1);
      events = observedevents;
run;
```

```
/*hospitalizations*/
data hosprate;
      set library.ratebase;
      keep wavetxt agegrp dosegrp nshhospcount pt;
      where wavetxt = "&currentwave" and agegrp in('<18 years', '18-49
years', '50-69 years', '70+ years');
      pt = vaccpt;
run;

proc sort data = hosprate; by dosegrp agegrp; run;

data provrate;
      set library.ratebase;
      keep wavetxt agegrp pt;
      where wavetxt = "&currentwave" and agegrp in('<18 years', '18-49
years', '50-69 years', '70+ years');
      pt = agept;
run;

proc sort data = provrate nodupkey; by agegrp; run;

ods graphics on;
ods select StdRate;
proc stdrate data=hosprate
      refdata=provrate
      method=direct
      stat=rate(mult=100000)
      plots=none;
      by dosegrp;
      population  event=nshhospcount total=pt;
      reference total=pt;
      strata agegrp;
      ods output StdRate=hospadj;
run;
ods graphics off;

data hospadj (keep=outcome dosegrp events cr ar);
      set hospadj;
      outcome = "Hospitalizations";
      cr = round(cruderate*365.5,0.1);
      ar = round(stdrate*365.5,0.1);
      events = observedevents;
run;

data library.vesummary;
      set hospadj deathadj;
run;
```